1976

A Computer Algorithm for the Natural Frequencies of Shafts

T. C. Scott

G. L. Davis

A COMPUTER ALGORITHM FOR THE NATURAL FREQUENCIES OF SHAFTS

T. C. SCOTT, SENIOR RESEARCH ENGINEER
G. L. DAVIS, MANAGER
ADVANCED ENGINEERING DEPARTMENT
SUNDSTRAND COMPRESSORS, BRISTOL, VIRGINIA

INTRODUCTION

Determination of the natural frequencies of compressor crankshafts requires the solution of a Holtzer-type multiple degree of freedom problem. For crankshafts, the number of different masses can become quite large so that a manual solution is extremely time consuming. While numerical methods are outlined in several texts [3]*, no simple computer program is available. This paper presents a FORTRAN subroutine for the solution of Holtzer-type problems along with several examples to illustrate its use.



1a) mass-spring system



1b) torsional system

FIG. 1: Holtzer-type Multiple Degree of Freedom Systems

* Numbers in [ ] are references at the end of this paper.

GOVERNING EQUATIONS

Figure (1) illustrates two common Holtzer-type multiple degree of freedom systems. The equations of motion for both systems are developed by displacing each mass from its equilibrium position. For the jth mass in the mass-spring system of Figure (1a)

$$m_j \frac{d^2x_j}{dt^2} = \sum F_j \tag{1}$$

while for the jth mass in the torsional system

$$J_j \frac{d^2\theta_j}{dt^2} = \sum \tau_j \tag{2}$$

The two systems are thus governed by the same equation so that we may develop the solution for either without loss of generality.

For free vibration of the system of n masses and n+1 springs the governing equations are

$$(j=1) \qquad m_1 \frac{d^2x_1}{dt^2} = -K_1 x_1 + K_2 [x_2 - x_1] \tag{3}$$

$$(j=2,\ldots,n-1) \qquad m_j \frac{d^2x_j}{dt^2} = -K_j [x_j - x_{j-1}] + K_{j+1} [x_{j+1} - x_j] \tag{4}$$

$$(j=n) \qquad m_n \frac{d^2x_n}{dt^2} = -K_n [x_n - x_{n-1}] - K_{n+1} x_n \tag{5}$$

The solution for each displacement has the form

$$x_j = A_j \sin \omega t \tag{6}$$

where A_j is the amplitude of vibration.

Substitution of this solution into the governing equations yields (j=1) $$-m_1\omega^2 A_1 = -K_1 A_1 + K_2[A_2 - A_1] \quad (7)$$

(j=2, . . . ,n-1) $$-m_j\omega^2 A_j = -K_j[A_j - A_{j-1}] + K_{j+1}[A_{j+1} - A_j] \quad (8)$$

(j=n) $$-m_n\omega^2 A_n = -K_n[A_n - A_{n-1}] - K_{n+1} A_n \quad (9)$$

One additional relation follows from applying the equation of motion to the entire system, namely,

$$\sum_{j=1}^{n} m_j \frac{d^2 x_j}{dt^2} = -K_1 x_1 - K_{n+1} x_n \quad (10)$$

whose solution using equation (6) is

$$-\omega^2 \sum_{j=1}^{n} m_j A_j = -K_1 A_1 - K_{n+1} A_n \quad (11)$$

The determination of the n natural frequencies $\omega_1, \omega_2, \ldots, \omega_n$ which satisfy equations (7), (8), (9), and (11) may be carried out by a number of numerical techniques. For our purposes here, the Holtzer method will be used since it is well known and easy to program for computer solution without the need of supporting routines for matrix manipulation. These factors along with the low machine computation times involved outweigh the efficiency advantage of more sophisticated schemes.

In the Holtzer method the amplitude A_1 is set at 1.0 then, for any guess of ω, equations (7) through (9) are solved sequentially for $A_2, A_3, \ldots, A_n$ These are then used to test for satisfaction of equation (11).

COMPUTER PROGRAM DESCRIPTION

The computer program searches for all natural frequencies which lie in the range

$$\omega_{min} \leq \omega \leq \omega_{max}$$

specified by the user. The approximate location of these frequencies is established by calculating

$$S = K_1 A_1 + K_{n+1} A_n - \omega^2 \sum_{j=1}^{n} m_j A_j \quad (12)$$

at $\omega_{min}, \omega_{min} + \Delta\omega, \omega_{min} + 2\Delta\omega, \ldots$ etc. where

$$\Delta\omega = \frac{\omega_{max} - \omega_{min}}{N_P} \quad (13)$$

N_P being the number of intervals specified by the user. This initial rough search can be quite coarse since its only purpose is to locate the natural frequencies between two local limits which lie between the adjoining maxima and minima of the S vs ω curve as shown in Figure (2). Each ω is then refined by iteration.



FIG. 2: Illustration of Solution Procedure

The computer program is presented in the form of a SUBROUTINE in FORTRAN-IV. It contains its own formats so that the user may have the results printed out if desired. The table below describes the arguments for the routine

```
SUBROUTINE HOLT(N,K,M,FMIN,FMAX,NP,EPS,A,
                FREQ,NF,IPRT)
```

FUNCTION: Natural frequencies by Holtzer method.

argument	input or output (I,O)	description
N	I	The number of masses.
K	I	An array dimensioned K(N+1) containing the spring constants.
M	I	An array dimensioned M(N) containing the masses.
FMIN	I	The lower limit of the search (Hz).
FMAX	I	The upper limit of the search (Hz).
NP	I	The number of intervals to use in the rough search.
EPS	I	The accuracy to which each frequency is to be found (Hz).
A	O	An array dimensioned A(N,N) to contain the relative amplitudes $\frac{A_1}{A_1}, \frac{A_2}{A_1}, \frac{A_3}{A_1}, \ldots$ at each natural frequency. The subscripting of this array is illustrated below.

argument	input or output (I,O)	description
		For example, A(3,4) means the relative amplitude of the 4th mass at the 3rd natural frequency.
FREQ	O	An array dimensioned FREQ(N) to contain the frequencies found (Hz).
NF	O	The number of frequencies found within the designated region of search.
IPRT	I	A print code which must be ≥0. If IPRT>2, no output will be printed by the subroutine. The effects of IPRT = 0, 1, or 2 are described below.

If IPRT is input as 0, the subroutine will print out the value of each natural frequency found in ascending order and the relative amplitude of each mass at each frequency. If the subroutine fails to find a particular frequency because it cannot converge to within the specified tolerance in 50 iterations, an error message will be printed.

If IPRT is input as 1, no output will be printed unless an error occurs in which case an error message will be written.

IF IPRT is input as 2, the results will be the same as for IPRT=0 with the addition that the values of ω and S will be printed at each point in the rough search. With this data the user can check to see that NP was large enough so that no frequencies in the interval were missed.

WHEN THE SUBROUTINE FAILS

An error can occur if NP is too small. In this case the iteration process will probably become unstable and result in a dump of the entire job.

The only other probable error comes from setting EPS too small so that the iteration cannot converge because round-off errors are bigger than EPS. In this case the routine will set the offending frequency equal to -1000 Hz as a signal to the calling program.

SUBROUTINE LISTING

The listing of the subroutine is presented in the left column of the following two pages.

M AND K UNITS

Any consistant system of mass and stiffness units with time in seconds will work.

EXAMPLE 1

As a simple example to illustrate the internal print features, consider the three mass system of Figure (3).



FIG. 3: Mass-Spring System for Example 1

In this case N=3 and since there is no fourth spring we have

M(1) = 4.0	K(1) = 3.0
M(2) = 2.0	K(2) = 1.0
M(3) = 1.0	K(3) = 1.0
	K(4) = 0.0

The dimensions of the M and K are assumed to be in consistant units by the computer routine so we shall not need them for this illustration.

Let us search over a range of 0.0 to 1.0 Hz with 20 steps and set IPRT=0 to get only the results printed. A MAIN program to do this is as follows:

```
DIMENSION K(4),M(3),FREQ(3),A(3,3)
REAL K,M
DATA K/3.,1.,1.,0./
DATA M/4.,2.,1./
CALL HOLT(3,K,M,0.,1.,20,.001,A,FREQ,NF,0)
STOP
END
```

Here we have set EPS at 0.001 Hz

The resulting output from the subroutine is reproduced in the right column of the following page.

(SUBROUTINE LISTING)

```
      SUBROUTINE HOLT(N,K,M,FMIN,FMAX,NP,EPS,A,FREQ,NF,IPRT)
C
C  NATURAL FREQUENCIES BY HOLTZER METHOD
C
      DIMENSION K(1),M(1),FREQ(1),A(N,1)
      REAL K,M
    4 FORMAT(1X,'---WARNING IN SUBROUTINE HOLT---',/,1X,
     1'CANNOT CONVERGE TO',I2,'TH FREQUENCY')
    5 FORMAT(1X,'---OUTPUT OF SUBROUTINE HOLT---')
    6 FORMAT(1X,'---RESULTS AT EACH FREQUENCY TRIED--',/,4X,
     1'HERTZ      SUM')
    7 FORMAT(/,1X,'--FINAL RESULTS, FREQUENCIES FOUND=',I4)
    8 FORMAT(1X,'AT',G11.4,' HERTZ,')
    9 FORMAT(1X,'A(',I2,')=',G11.4,3X,'A(',I2,')=',G11.4,3X,
     1'A(',I2,')=',G11.4)
   10 FORMAT(1X,2G11.4)
C  SET LIMITS
      PI2=6.283185307
      E=EPS*PI2
      DW=PI2*(FMAX-FMIN)/NP
      J=0
      WMI=PI2*FMIN
      IF(WMI.EQ.0.) WMI=DW/100.
      IF(IPRT.EQ.2) WRITE(3,5)
      IF(IPRT.EQ.0) WRITE(3,5)
      W2=WMI*WMI
      SUM=M(1)*W2-K(1)
      B=1.-SUM/K(2)
      DO 21 II=3,N
      SUM=SUM+M(II-1)*W2*B
   21 B=B-SUM/K(II)
      SMI=SUM+(M(N)*W2-K(N+1))*B
      IF(IPRT.EQ.2) WRITE(3,6)
      IF(IPRT.NE.2) GO TO 11
      FF=WMI/PI2
      WRITE(3,10) FF,SMI
   11 DO 100 I=1,NP
      WMO=WMI+DW
      W2=WMO*WMO
      SUM=M(1)*W2-K(1)
      B=1.-SUM/K(2)
      DO 31 II=3,N
      SUM=SUM+M(II-1)*W2*B
   31 B=B-SUM/K(II)
      SMO=SUM+(M(N)*W2-K(N+1))*B
      IF(IPRT.NE.2) GO TO 12
      FF=WMO/PI2
      WRITE(3,10) FF,SMO
   12 IF(SMI*SMO.GE.0.) GO TO 50
C  SEARCH FOR FREQUENCY
      J=J+1
      WMIN=WMI
      WMAX=WMO
      SMIN=SMI
      SMAX=SMO
      DO 40 KK=1,50
      W=(WMIN+WMAX)/2.
      W2=W*W
      SUM=M(1)*W2-K(1)
      B=1.-SUM/K(2)
      DO 37 II=3,N
      SUM=SUM+M(II-1)*W2*B
```

(COMPUTER OUTPUT FROM EXAMPLE 1)

```
---OUTPUT OF SUBROUTINE HOLT---

--FINAL RESULTS, FREQUENCIES FOUND=   3
AT 0.7589E-01 HERTZ,
A( 1)= 1.000  A( 2)= 4.091  A( 3)= 5.321
AT 0.1743     HERTZ,
A( 1)= 1.000  A( 2)= 0.201  A( 3)=-1.080
AT 0.2150     HERTZ,
A( 1)= 1.000  A( 2)=-2.296  A( 3)= 2.785
```

If the results from the rough search are also desired, re-running the program with IPRT=2 gives the following output

```
---OUTPUT OF SUBROUTINE HOLT---
--RESULTS AT EACH FREQUENCY TRIED--
    HERTZ      SUM
 0.5000E-03 -4.000
 0.5050E-01 -1.940
 0.1005      1.544
 0.1505      1.497
 0.2005     -1.006
 0.2505      15.04
 0.3005      110.0
 0.3505      408.6
 0.4005      1130.
 0.4505      2626.
 0.5005      5418.
 0.5505     0.1025E 05
 0.6005     0.1813E 05
 0.6505     0.3042E 05
 0.7005     0.4884E 05
 0.7505     0.7561E 05
 0.8005     0.1135E 06
 0.8505     0.1657E 06
 0.9005     0.2365E 06
 0.9505     0.3307E 06
  1.000     0.4540E 06
```

in addition to the previous output shown for the IPRT=0 case. The reader will note that the first frequency of the rough search is not quite 0.0 as specified by FMIN. This is because ω = 0.0 is the trivial solution in some cases so that

```
 37  B=B-SUM/K(II)
     SUM=SUM+(M(N)*W2-K(N+1))*B
     ER=ABS(WMAX)-ABS(WMIN)
     IF(ABS(ER).LE.E) GO TO 45
     IF(SMIN*SUM.GT.0.) GO TO 35
     SMAX=SUM
     WMAX=W
     GO TO 40
 35  SMIN=SUM
     WMIN=W
 40  CONTINUE
C  FAILURE TO CONVERGE
     FREQ(J)=-1000.
     IF(IPRT.GT.2) GO TO 50
     WRITE(3,4) J
     GO TO 50
 45  FREQ(J)=W/PI2
 50  WMI=WMO
100  SMI=SMO
C  COMPUTE AMPLITUDES
     NF=J
     IF(IPRT.EQ.0) WRITE(3,7) NF
     IF(IPRT.EQ.2) WRITE(3,7) NF
     IF(NF.LE.0) RETURN
150  DO 200 I=1,NF
     IF(FREQ(I).LT.-900.) GO TO 200
     W=FREQ(I)*PI2
     W2=W*W
     A(I,1)=1.
     SUM=M(1)*W2-K(1)
     A(I,2)=A(I,1)-SUM/K(2)
     DO 180 J=3,N
     SUM=SUM+M(J-1)*W2*A(I,J-1)
180  A(I,J)=A(I,J-1)-SUM/K(J)
     IF(IPRT.GT.2) GO TO 200
     IF(IPRT.EQ.1) GO TO 200
     WRITE(3,8) FREQ(I)
     WRITE(3,9) (J,A(I,J),J=1,N)
200  CONTINUE
     RETURN
     END
```

(EXAMPLE 1 CONT'D)

when FMIN is set at 0.0, the routine always starts the search a small step away.

In analysing general problems like this one, the Holtzer method is at a disadvantage since one may not know in advance what range the natural frequencies lie in. The subroutine is better suited for crankshaft frequencies where the designer generally knows what range of frequencies will be critical and is only interested in those natural frequencies falling in this range.

EXAMPLE 2

As a second example, consider the crankshaft of a hermetic refrigeration compressor shown in Figure (4). This system has been modeled by nine masses with polar moments of inertia J_1 to J_9 and associated torsional stiffnesses as listed in Table (1). The methods for determining these numerical values are outlined in several texts [1],[2] and will not be discussed here.

For this system, the J values go in the M array of the subroutine and, since the shaft is not constrained at either end,

K(1)= K(10)= 0.0

For a typical compressor running at 3600 RPM, the designer is only concerned with shaft natural frequencies below, say, several multiples of the shaft speed. We shall search in a somewhat extended range of 0.0 to 10000. Hz using 100 steps. A MAIN program for this is as follows

```
 DIMENSION K(10),M(9),FREQ(9),A(9,9)
 REAL K,M
 DATA K/0.,2.86E5,1.29E6,3.43E6,1.2E6,
A1.2E6,1.2E6,1.4E6,2.07E6,0./
 DATA M/.0213,3.1E-4,1.97E-4,1.7E-3,
A9.3E-5,9.3E-5,1.7E-3,5.66E-4,1.88E-4/
 CALL HOLT(9,K,M,0.,10000.,100,.1,A,FREQ,
ANF,0)
 STOP
 END
```

Here, EPS is set at 0.1 Hz and IPRT=0
The resulting output is as follows

```
 ---OUTPUT OF SUBROUTINE HOLT---

 --FINAL RESULTS, FREQUENCIES FOUND=  3
 AT  1102.      HERTZ,
 A( 1)=  1.000  A( 2)=-2.572   A( 3)=-3.334
 A( 4)=-3.612   A( 5)=-4.160   A( 6)=-4.692
 A( 7)=-5.207   A( 8)=-5.345   A( 9)=-5.368
 AT 3211.       HERTZ,
 A( 1)=  1.000  A( 2)=-29.31   A( 3)=-33.16
 A( 4)=-33.83   A( 5)=-16.26   A( 6)= 1.836
 A( 7)= 19.87   A( 8)= 25.51   A( 9)= 26.48
 AT 8122.       HERTZ,
 A( 1)=  1.000  A( 2)=-192.9   A( 3)=-115.2
 A( 4)=-68.78   A( 5)= 317.7   A( 6)= 639.9
 A( 7)= 833.0   A( 8)=-1636.   A( 9)=-2141.
```

The remaining natural frequencies all lie above 10000. Hz. The lowest natural frequency of 1102. Hz is considerably higher than the shaft speed and should cause no problems.



FIG. 4: Hermetic Compressor Crankshaft for Example 2

CONCLUSIONS

A FORTRAN subroutine has been presented for the solution of Holtzer-type vibration problems. The routine is easy to use and contains its own output formats. Natural frequencies of compressor crankshafts is one of the problems for which this routine is well suited.

NOMENCLATURE

- A = Amplitude
- J = Moment of inertia
- K = Stiffness
- m = Mass
- t = Time
- ω = Frequency
- X = Linear displacement
- θ = Angular displacement

REFERENCES

1. Den Hartog, J.P., Mechanical Vibrations 3rd ed., McGraw-Hill, 1947
2. Shigley, J.E., Theory of Machines, McGraw-Hill, 1961
3. Thomson, W.T., Vibration Theory and Applications, Prentice-Hall, 1965

TABLE 1: Data for Figure (4)

J (in-lbf-sec^2)	K (in-lbf)
J_1 = .0213	K_2 = 2.86×10^5
J_2 = 3.1×10^{-4}	K_3 = 1.29×10^6
J_3 = 1.97×10^{-4}	K_4 = 3.43×10^6
J_4 = 1.7×10^{-3}	K_5 = 1.2×10^6
J_5 = 9.3×10^{-5}	K_6 = 1.2×10^6
J_6 = 9.3×10^{-5}	K_7 = 1.2×10^6
J_7 = 1.7×10^{-3}	K_8 = 1.4×10^6
J_8 = 5.66×10^{-4}	K_9 = 2.07×10^6
J_9 = 1.88×10^{-4}	